

ApptList

All the appointments in the control.

Usage

```
Set_Property(CtrlEntID, "OLE.ApptList", Appts)
```

Values

@FM delimited list of appointments.

Default: ""

Remarks

The ApptList property defines all the appointments for the control. This property is an @FM delimited array, with each field containing these values:

| Position | Name | Type | Description |
|----------|-----------------|----------|--|
| <x, 1> | Calendar Key | Text | The unique key of the calendar to contain this appointment. |
| <x, 2> | Key | Text | The unique key for this appointment. No two appointments can have the same key, even if they are in different calendars. |
| <x, 3> | Start | DateTime | The date and time, in OI internal format, that the appointment begins. |
| <x, 4> | End | DateTime | The date and time, in OI internal format, that the appointment ends. |
| <x, 5> | Subject | Text | The short summary describing the appointment. |
| <x, 6> | Body | Text | The details of the appointment. |
| <x, 7> | Location | Text | The brief description of the appointment's location. |
| <x, 8> | Label Key | Text | The unique key of a previously defined label, which will categorize and colorize this appointment. |
| <x, 9> | Busy Status | Option | One of four options: "Free", "Tentative", "Busy", or "OutOfOffice". |
| <x, 10> | Importance | Option | One of three options: "Low", "Normal", or "High". |
| <x, 11> | Is Meeting Flag | Boolean | Determines if the appointment is a meeting. |
| <x, 12> | Is Private Flag | Boolean | Determines if the appointment is private. |

Example

```

// Define labels
Labels = ""
Labels<-1> = "Personal" :@VM:"LightBlue"
Labels<-1> = "Urgent" :@VM:"LightRed"
Set_Property(@Window:".OLE_CALENDAR", "OLE.Labels", Labels)

Appts = ""

// A personal appointment to remember drycleaning
Appts<1, 1> = "PERSONAL"
Appts<1, 2> = "10001"
Appts<1, 3> = IConv(OConv(Date(), "D4/"):" 8:00AM", "DT")
Appts<1, 4> = IConv(OConv(Date(), "D4/"):" 8:30AM", "DT")
Appts<1, 5> = "Drycleaning"
Appts<1, 6> = "Don't forget to pick up dry cleaning on the way to work."
Appts<1, 7> = "Main St."
Appts<1, 8> = "Personal"
Appts<1, 9> = "OutOfOffice"
Appts<1, 10> = "High"

// A work meeting
Appts<2, 1> = "WORK"
Appts<2, 2> = "10002"
Appts<2, 3> = IConv(OConv(Date(), "D4/"):" 1:00PM", "DT")
Appts<2, 4> = IConv(OConv(Date(), "D4/"):" 3:00PM", "DT")
Appts<2, 5> = "Board Meeting"
Appts<2, 6> = "Meeting with the Board to discuss next fiscal quarter."
Appts<2, 7> = "Conference Call"
Appts<2, 8> = "Urgent"
Appts<2, 10> = "High"
Appts<2, 11> = 1

Set_Property(@Window:".OLE_CALENDAR", "OLE.ApptList", Appts)

```